

組織内ソフトとパーソナルソフトの有効利用について*

An available using of both systemic soft and personal soft

岩佐 哲夫¹⁾

Tetsuo IWASA

宮井 義裕²⁾

Yoshihiro MIYAI

小林 信雄³⁾

Nobuo KOBAYASHI

Recently micro-computers called as personal computer or office computer are generalized rapidly, so that office automation (OA) is pushing forward in various fields. The other hand, the big computers are indispensable to the company of business and micro-computers tend to be connected with big computers. But many users operate the micro-computer and the big computer as off line now. In such situation, this paper proposes an available using of soft between big computers and micro computers by through medium of OCR (Optical Character Reader), code converter and etc.

The contents of this paper investigated effects of available usage.

1. ま え が き

小型電子計算機のめざましい発達は大形電子計算機の動向をも左右し、エレクトロニクスの発達と相俟って、正に情報化時代の一翼を担いつつある。小型電子計算機の普及の特徴は、大形機が特定の業種を中心に普及したのと反対に、個人の利用を中心にしてOA, FAからホビーに至る幅の広さと年令層の厚さにあるといえる。勿論、価格的にも一個人の手の届く程度であることも家電製品並の普及を実現した大きな要因である。

しかし、従来の家電製品と異なる点は、その用途が限定されていないことである。このことは我々が想像もしなかった新しい利用や活用が期待される。その意味では、小型電子計算機が幾多の変遷をするものと思われる。その兆候の一つとして、現在の通信回線を介して行なわれているモデムやカプラーに代って光通信による大形機との接続が話題になっている。この方法は多くの小型電子計算機を大形機と接続することにより、従来のTSSとは異なった利用方法を可能にする。すなわち、小型機側にも処理能力を持つため大形機オンリーではなく、業務の内容によって、その使い分けが可能になる。このような集中管理と分散管理の使い分けは今後の主流になるものと考えられる。このような使用形態をオフラインでも可能に出来れば組織として繋がり無し分野での共同作業、あるいは経費的、能力的な面でオンラインの遅れている分野でも、従来よりはるかに効果的な利用が期待できる。本研究はOCR、小型電子計算機の端末機器を活用して、組織内ソフトとパーソナルソフトの効果的な利用を、実作業用のプログラムを開発し実践的な検討を行なったものである。

2. 機器構成と有効利用

前述のように大形機にパソコンが接続できる状態が可能であれば、組織内のソフトとパーソナルソフトをより効果的に利用できることは言うまでもない。ここで言う組織内ソフトとは組織として共通の目的のために用意されたソフトを意味し、パーソナルソフトとは個人あるいは特定の部署のみが管

*昭和59年4月7日原稿受理

1), 2), 3) 大阪産業大学工学部

理するソフトを意味している。今節で述べる組織内ソフトとパーソナルソフトの関係とは大型機とパソコンがオフラインの状態、すなわち使用言語もコードも現状のままであることを意味し、その目的は両者に互換性あるいは分担業務を可能にするシステム設計をすることである。Fig.1に有効利用に関係する機器構成を示す。従来の使用は②—①のバッチ処理、③—①のTSS処理が主体で④—①のフロッピーからの利用も一部行なわれている。なお本学ではOCR本体にカードせん孔機、フロッピーディスクを装備し、バッチ処理としてのカード、ディスクットの作成はすべてOCRで行なっている。また簡単なプリンタ、カード読取機も付設しており、ディバッグ、ソースリストの作成も可能である。OCRの利用に関しては月刊雑誌「事務管理」（1982年6月号・日刊工業新聞社発行）に「大阪産業大学におけるOCRの活用」でも紹介したように、入力手段として今後ますます普及するものと考えられる。⑥のパソコンにはASCIIコードとIBMコードの変換システムを用意し、大型機とパソコンのフロッピーの互換性を持たせている。




Fig.1 System of equipments.

このように⑤のOCR、あるいは⑥のパソコンとコード変換のソフトを用意することによって、業務の質的改善はもちろん従来の大型機だけ、あるいはパソコンだけという利用形式から、両者の長を生かした有効利用が期待できる。その具体例は次節に紹介するが、その主な活用方法は下記の通りである。(1)マークカードの活用 マーク入力によるアンケート調査、選択肢試験は広く行なわれている。大型機あるいはパソコンのマークカード読取機から入力したデータを④、⑥のフロッピーを介してデータを共用し、その用途に応じて処理を行なう。(2)OCRシートの活用 手書き文字による入力を⑤のOCRでフロッピーに落し、その用途によって使い分ける。(3)サテライト的活用 大型機あるいはパソコンで処理した結果をフロッピーを介して再処理を行なう。(4)その他の活用 パソコンで最も広く利用されているBASIC言語の他に種々のOSが開発されており、これらOSとの互換機能を果たす役割も期待できる。

3. システム開発例

大阪産業大学産業研究所電子計算研究室の研究活動の一環としてCAI研究会を開催しており、その研究会を遂行していく過程で本来の研究を進める以前の問題として解決しておかねばならない教育事務上のプログラムがあった。その中から2, 3の例を紹介する。

3.1 汎用アンケートシステム

教育機関に限らず、アンケート調査は判断、対策等の情報を得る手段として広く行なわれている。その集計方法にはいろいろあるが、マークカードを用いて母集団全体にアンケートを行ったり、それと同等の調査を行なう場合、その度にプログラムを作成することなく集計作業をやりたい。すなわち、プログラムを知らない担当者でも気軽にアンケート調査ができることを目的としてプログラムを作成した。以下にその内容を説明する。

学 生 番 号	問 題 番 号	問 題															
		1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
1	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
2	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
3	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
4	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
5	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
6	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
7	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
8	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
9	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
10	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
11	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
12	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
13	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
14	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
15	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○
16	1	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○	○

Fig.2 Mark card

数, 男子回答者数, 女子回答者数とそのパーセントを出力する。

次にAND, OR のある指示 (入力) についても同様の出力をする。

AND とはある質問に対し○番と回答し, しかも他の質問に対して×番と回答したもの。OR とはある質問に対し○番と回答, あるいは他の質問に対して×番と回答したもの。というように相互関連の事象を集計する。

(3) 処 理

大型機で処理する場合はマークカードの後に/*カードを入れ, その後にAND, OR の集計指示カードをつける。(使用言語 FORTRAN)

パソコンの場合マークカードをフロッピーに落とし, ASCII コードに変換した後に処理を行なう。ただし, AND, OR の指示はキー入力で行なう。Fig.3 に大型機でAND, OR の指示を行なう場合のデータ FORMAT を示す。

このように BASIC, FORTRAN の両言語で同一プログラムを作成し, 入力データを共通にすれば対象者数, 集計方法, 時間, 追加あるいは検討のための処理等を配慮して担当者が都合のよい方を選んで集計することができる。

Fig.4 に FORTRAN プログラム, Fig.5 に BASIC プログラム, Fig.6 にアンケート出力結果の一部を示す。なお, BASIC プログラムは PC-8801 を対象機種とした。

Fig.4 Opinion survey program by FORTRAN

SOURCE STATEMENT

```

DIMENSION MD(700,33),LTL(10)
NO=0
NB=0
NG=0
DO 10 I=1,700
  READ(5,500,END=20) MD(I,33),(MD(I,J),J=1,32)
500 FORMAT (9X,33I1)
  NO=NO+1
  IF(MD(I,33).EQ.1) GO TO 10
  NG=NG+1
10 CONTINUE
20 NB=NO-NG


```

(1) 概 要

Fig.2 は後述のマークテストの回答カードである。このマークカードを流用して, 1Aから16Bまでの最大32問の質問ができる。選択肢は1問につき1~9の番号を選ぶ, ただし同一の質問に対して複数個の回答をさせてはならない。また回数欄は男子1, 女子2を記入させる。

(2) 機 能

各質問に対して各々の選択肢ごとに全回答者



例

Q	A	a	B	b
2	3	1	4	3

Q : AND=2, OR=3
A : 質問番号, a : 選択番号
B : 質問番号, b : 選択番号

例 : 質問3で1と答え, しかも
質問4で3と答えたもの。

Fig.3 Data format

```

      READ(5,499) MON,IY,IM,ID,(LTL(I),I=1,10)
      WRITE(6,600) NO,NB,NG,IY,IM,ID,(LTL(I),I=1,10)
499  FORMAT(4I2,10A4)
600  FORMAT(1H1,5Hアキテ,2X,4Hソウズ,15,2X,6Hタ*ンシズ,15,
6     4Hシ*ヨシ,15,3X,3(I2,1H,),3X,10A4)
      WRITE(6,609)
609  FORMAT(1H ,3HNO.,,1X,1HQ,3X,1HA,2X,1HQ,2X,1HA,2X,
6     4Hソウズ,3X,1H%,4X,4Hタ*ンシ,3X,1H%,4X,4Hシ*ヨシ,3X,1H%)
      LL=0
30   READ (5,501,END=200) N,NQ,NA,MQ,MA
501  FORMAT(I1,I2,I1,I2,I1)
      LL=LL+1
      KT=0
      KB=0
      KG=0
      IF(LL.EQ.1) GO TO 100
      IF(MOD(LL,50)-1.EQ.0) WRITE(6,605)
      IF(MOD(LL,5)-1.EQ.0) WRITE(6,608)
100  IF(N.NE.2) GO TO 150
C ----- N = 2 / ハ*アイ -----
      DO 120 I=1,NO
      IF(MD(I,NQ).EQ.NA.AND.MD(I,MQ).EQ.MA) GO TO 110
      GO TO 120
110  KT=KT+1
      IF(MD(I,33).EQ.2) KG=KG+1
120  CONTINUE
      KB=KT-KG
      PT=FLOAT(KT)/FLOAT(NO)*100.0
      PB=FLOAT(KB)/FLOAT(NB)*100.0
      PG=FLOAT(KG)/FLOAT(NG)*100.0
      WRITE(6,602) N,NQ,NA,MQ,MA,KT,PT,KB,PB,KG,PG
602  FORMAT(1H ,I3,4I3,3(I6,F6.1))
      GO TO 30
150  IF(N.NE.3) GO TO 300
C ----- N = 3 / ハ*アイ -----
      DO 180 I=1,NO
      IF(MD(I,NQ).EQ.NA.OR.MD(I,MQ).EQ.MA) GO TO 160
      GO TO 180
160  KT=KT+1
      IF(MD(I,33).EQ.2) KG=KG+1
180  CONTINUE
      KB=KT-KG
      PT=FLOAT(KT)/FLOAT(NO)*100.0
      PB=FLOAT(KB)/FLOAT(NB)*100.0
      PG=FLOAT(KG)/FLOAT(NG)*100.0
      WRITE(6,602) N,NQ,NA,MQ,MA,KT,PT,KB,PB,KG,PG
      GO TO 30
300  WRITE(6,610) N
610  FORMAT(1H ,I3,9Hア*マ チイツク)
      GO TO 30
605  FORMAT(1H1,3HNO.,,1X,1HQ,3X,1HA,2X,1HQ,2X,1HA,2X,
6     4Hソウズ,3X,1H%,4X,4Hタ*ンシ,3X,1H%,4X,4Hシ*ヨシ,3X,1H%)
608  FORMAT(1H )
200  CALL LST(MD,NQ,NB,NG,MON)
      STOP
      END

```

```

SUBROUTINE LST(MD,NO,NB,NG,MON)
DIMENSION MD(700,33),NT(3,9,32),P(3,9,32),T(3),NN(9)
DO 5 I=1,9
5 NN(I)=I
DO 10 I=1,2
DO 10 J=1,9
DO 10 K=1,MON
NT(I,J,K)=0
10 CONTINUE
C-----NT(1-BOY --- NT(2---GIRL --- NT(3--TOTAL -----
DO 20 I=1,NO
L=1
IF(MD(I,33).EQ.2) L=2
DO 30 K=1,MON
N=MD(I,K)
IF(N.EQ.0) GO TO 30
NT(L,N,K)=NT(L,N,K)+1
30 CONTINUE
20 CONTINUE
DO 40 I=1,9
DO 40 J=1,MON
NT(3,I,J)=NT(1,I,J)+NT(2,I,J)
40 CONTINUE
T(3)=NO
T(2)=NG
T(1)=NB
DO 50 I=1,3
DO 50 J=1,9
DO 50 K=1,MON
P(I,J,K)=FLOAT(NT(I,J,K))/T(I)*100.0
50 CONTINUE
DO 60 K=1,MON
IF(MOD(K,10)-1.EQ.0) WRITE(6,600) (NN(I),I=1,9)
WRITE(6,601) (NT(1,J,K),P(1,J,K),J=1,9)
WRITE(6,602) K,(NT(2,J,K),P(2,J,K),J=1,9)
WRITE(6,603) (NT(3,J,K),P(3,J,K),J=1,9)
WRITE(6,605)
60 CONTINUE
605 FORMAT(1H )
600 FORMAT(1H1,8X,9(3X,2Hニ, I2,4X,1H%))
601 FORMAT(1H ,5X,4H? *ニ,9(I5,F7.1))
602 FORMAT(1H ,I3,2X,4H? *ニ,9(I5,F7.1))
603 FORMAT(1H ,4X,5H? *ウケイ,9(I5,F7.1))
RETURN
END

```

Fig.5 Opinion survey program by BASIC

```

100 'SAVE"アンケート 88".A
110 '.....
120 '.....
130 '..... アンケート シュウケイ .....
140 '.....
150 '.....
160 ' DATA NAME
170 PRINT CHR$(12)::PRINT TAB(20)"DATA FILE NAME ";;INPUT ;DATO$ :WA=7:YE=6
180 *HAZIME
190 WIDTH 80,25
200 CONSOLE,,0,1
210 PRINT CHR$(12) :COLOR 2
220 LOCATE 26,2:PRINT "r";STRING$(13,"-");"r"
230 LOCATE 26,3:PRINT "| アンケート シュウケイ |"
240 LOCATE 26,4:PRINT "L";STRING$(13,"-");"L"
250 COLOR WA:LOCATE 21,7:PRINT "1) センタイノ シュウケイ "
260 COLOR WA:LOCATE 21,9:PRINT "2) (AND/OR)ニ ヨル シュウケイ"
270 COLOR YE:LOCATE 21,11:PRINT "3) ショト シュウリョウ "
280 COLOR 4:LOCATE 29,13:INPUT "No";N
290 IF N<1 OR N>3 THEN PRINT TAB(20)"マチカイ":GOTO 280
300 ON N GOTO *ZENTAI,*SIKAMO.MATAWA
310 CONSOLE,,1,0
320 PRINT CHR$(12):END
330 '.....
340 '◆◆ センタイノ アンケートヲ シュウケイ サセル フォグラム ◆◆
350 '.....
360 *ZENTAI
370 CONSOLE,,0,1
380 WIDTH 80,25:PRINT CHR$(12)
390 NO=0:NB=0
400 DIM S(3,33,9),P(3,33,9),PP(3)
410 COLOR WA:PRINT TAB(10)"シハラク オマチクダサイ! イマ アンケート テータヲ ヨンテ イマス!":PRINT
420 OPEN "2:"+DATO$ FOR INPUT AS #2
430 IF EOF(2) THEN *TOZIRU
440 LINE INPUT #2,A$
450 I=VAL(LEFT$(A$,1))
460 IF I=1 THEN NB=NB+1
470 FOR J=2 TO 33:IF J=33 THEN A=A+1:LOCATE 20,5 :COLOR YE:PRINT USING"### ニンメ"
:A
480 K=VAL(MID$(A$,J,1))
490 IF K=0 THEN 510
500 S(I,J,K)=S(I,J,K)+1
510 NEXT J
520 NO=NO+1
530 GOTO 430
540 *TOZIRU :COLOR WA
550 CLOSE #2 :PRINT TAB(20)"DATA READ END"
560 NG=NO-NB:PRINT TAB(10)"ソウスウ";NO;"タンシ";NB"シヨシ";NG
570 PB=NB/NO*100:PB=INT(PB+.5)
580 PG=NG/NO*100:PG=INT(PG+.5)
590 PRINT:PRINT:PRINT TAB(10)"イマ アンケートヲ シュウケイシテ イマス! 32モン":PRINT
600 FOR J=2 TO 33:LOCATE 20,15:COLOR YE:PRINT USING"## モン";J-1;
610 FOR K=1 TO 9
620 S(3,J,K)=S(1,J,K)+S(2,J,K)
630 P(3,J,K)=INT(S(3,J,K)/NO*100+.5)
640 P(1,J,K)=INT(S(1,J,K)/NB*100+.5)
650 P(2,J,K)=INT(S(2,J,K)/NG*100+.5)
660 NEXT K
670 NEXT J :GOSUB *COPYNO
680 *AGIN
690 COLOR 2:PRINT:PRINT:PRINT "LPRINT チュウ!":COLOR WA
700 FOR IY=1 TO COPP
710 LPRINT CHR$(14)" センタイノ アンケート シュウケイ ヒョウ":LPRINT :LPRINT CHR$(15):LPRINT
:LPRINT CHR$(27);"B";
720 LPRINT :LPRINT :LPRINT TAB(10)
730 LPRINT USING "アンケート ソウスウ ###ニン (タンシ ###ニン ###%";NO,NB,PB;
740 LPRINT USING " シヨシ ###ニン ###%";NG,PG :LPRINT
750 X=0
760 LPRINT TAB(8)"r";:C$="r":GOSUB *RAIN
770 FOR I=1 TO 9
780 LPRINT TAB(I*8) USING "| #ハン";I;

```



```

1500 INPUT #1,A$
1510 NO=NO+1
1520 IF VAL(LEFT$(A$,1))=1 THEN NB=NB+1
1530 B$=MID$(A$,A1+1,1)
1540 C$=MID$(A$,B1+1,1)
1550 IF NQ=1 THEN GOTO *SIKAMO ELSE IF NQ=2 THEN GOTO *MATAWA
1560 GOTO 1490
1570 *SIKAMO
1580 IF A2=VAL(B$) AND B2=VAL(C$) THEN *NT '◆◆◆◆◆◆ AND ノ ハ`アイ ◆◆◆◆◆◆
1590 GOTO 1490
1600 *MATAWA
1610 IF A2=VAL(B$) OR B2=VAL(C$) THEN *NT '◆◆◆◆◆◆ O R ノ ハ`アイ ◆◆◆◆◆◆
1620 GOTO 1490
1630 *JAMP
1640 GOTO 1490
1650 *NT
1660 NT=NT+1
1670 IF VAL(LEFT$(A$,1))=1 THEN BT=BT+1
1680 GOTO *OWARI
1690 *SIMARU
1700 CLOSE #1
1710 NG=NO-NB
1720 GT=NT-BT
1730 PN=NT/NO*100:PN=INT(PN+.5)
1740 PB=BT/NB*100:PB=INT(PB+.5)
1750 PG=GT/NG*100:PG=INT(PG+.5)
1760 PRINT
1770 IF NQ=1 THEN D$="シカモ":E$="and"
1780 IF NQ=2 THEN D$="マタハ":E$="o r"
1790 COLOR 2:LOCATE 14,17:PRINT A1;"モンメヲ";A2;"ト コタエ (" :D$;")";B1;"モンメヲ";B2;"ト コ
タエ ヒト"
1800 COLOR WA:PRINT
1810 V$=" タ`ンシハ":F$=" シ`ヨシハ":O$="コ`ウケイハ"
1820 PRINT TAB(19) USING "& &.....###/###ニ`ン ###%";V$,BT,NB,PB
1830 PRINT TAB(19) USING "& &.....###/###ニ`ン ###%";F$,GT,NG,PG
1840 PRINT TAB(19) USING "& &.....###/###ニ`ン ###%";O$,NT,NO,PN
1850 COLOR YE:LOCATE 28,22:INPUT "PRINT シマスカ Yes=1.No=2";AAA :IF AAA =2 THEN 2
050
1860 LOCATE 28,23:PRINT "LPRINT チュウ!" '28,23
1870 IF Q=1 OR Q=2 THEN *KOMENDO
1880 LPRINT " AND,OR ニ ヨル アンケート シュウケイ"
1890 LPRINT TAB(27)"r";:X=16:V=2:C$="r":P$="r":GOSUB *SEN
1900 W$="ソウズウ":R$="タ`ンシ":H$="シ`ヨシ"
1910 LPRINT TAB(26)
1920 LPRINT USING " | & (#ニ`ン)":W$,NO,R$,NB,H$,NG;:LPRINT " |"
1930 LPRINT "r";:X=26:V=3:C$="+":P$="r":GOSUB *SEN
1940 *KOMENDO
1950 LPRINT USING " | ## ハ`ンノ # (& &)## ハ`ンノ #":A1,A2,E$,B1,B2;
1960 LPRINT USING " | ###ニ`ン ###% ":NT,PN,BT,PB,GT,PG;:LPRINT " |"
1970 LPRINT "L";:X=26:V=3:C$="+":P$="L":GOSUB *SEN
1980 GOTO 2050
1990 *SEN
2000 LPRINT STRING$(X,"-");
2010 FOR I=1 TO V
2020 LPRINT C$:STRING$(16,"-");
2030 NEXT I:LPRINT P$
2040 RETURN
2050 COLOR YE:LOCATE 15,23:INPUT "マタ`ヤリマスカ? AND=1 OR=2 END=3 ":Q
2060 IF Q<1 OR Q>4 THEN 2050
2070 IF Q=1 OR Q=2 THEN *SIKAMO.MATAWA ELSE IF Q=3 THEN *HAZIME
2080 *COPYNO
2090 COLOR 2:PRINT :LOCATE 20,20:INPUT "COPY マイスク":COPP
2100 RETURN
2110 END

```


Fig.6 A example of question and output

- 13 A. あなたはいつも補聴器をかけていますか。
 (1)毎日かけている (2)ときどき (3)あまりかけていない
- 13 B. 補聴器は役に立っていますか。
 (1)非常に役立っている (2)あまり役に立たない
- 14 A. 補聴器をかけたら人の話している声や電話の声がよくわかりますか。
 (1)よくわかる (2)聞きとりにくい (3)ぜんぜんわからない

セマンタイ ノ アンケート シュウケイ ヒヨウ

アンケート ソウズウ 87ニン (タンシ 53ニン 61% ショシ 34ニン 39%)

		1ハン		2ハン		3ハン		4ハン		5ハン		6ハン	
		ニン	%	ニン	%	ニン	%	ニン	%	ニン	%	ニン	%
1	タンシ	15	28	10	19	11	21	11	21	6	11	0	
	ショシ	11	32	8	24	7	21	5	15	3	9	0	
	ゴウケイ	26	30	18	21	18	21	16	18	9	10	0	
2	タンシ	27	51	19	36	5	9	1	2	0	0	0	
	ショシ	12	35	20	59	2	6	0	0	0	0	0	
	ゴウケイ	39	45	39	45	7	8	1	1	0	0	0	6
3	タンシ	15	28	29	55	8	15	0	0	0	0	0	0
	ショシ	12	35	18	53	3	9	0	0	0	0	0	0
	ゴウケイ	27	31	47	54	11	13	0	0	0	0	0	0
4	タンシ	44	83	8	15	1	2						
	ショシ	31	91	2	6	0							
	ゴウケイ	75	86	10	11								

AND, OR ニヨル アンケート シュウケイ

	ソウズウ (87ニン)	タンシ (53ニン)	ショシ (34ニン)
1ハンノ2 (and) 2ハンノ2	7ニン 8%	4ニン 8%	3ニン 9%
2ハンノ2 (and) 3ハンノ1	15ニン 17%	6ニン 11%	9ニン 26%
1ハンノ2 (or) 2ハンノ2	50ニン 57%	25ニン 47%	25ニン 74%
3ハンノ1 (or) 4ハンノ2	36ニン 41%	22ニン 42%	14ニン 41%
2ハンノ2 (or) 2ハンノ3	46ニン 53%	24ニン 45%	22ニン 65%

3. 2 個人研究図書 of 検索・集計システム

図書館に所属しないけれども部門、個人が管理している文献、書籍、資料等を掌握しておかねばならない場合は多い。当学の産業研究所では各教員が一定枠内で購入する研究図書の台帳作成、予算処理を行なっている。当初はタナックシステム（カードによるソート、マージ）を採用し、一応の成果をあげていたが年数の経過と共に冊数も増加し、事務担当者の業務量が年々増加している。このことは、我々教員にとってサービスの低下を招くだけでなく教育研究活動に支障を来す恐れもある。しかしながら、この業務は事務所の所轄であり、我々教員ができるものではない。たまたま、産業研究所の事務長からパソコン程度で事務処理ができないものかと相談を受け、それではテスト的にやってみましょうか……という経緯で作成したのが本システムである。テスト的という意味は、本来組織とし

て運用されるべきであるから、システムの採否は事務室にお任せするということである。以下にシステムの概要と入出力について説明する。


(1) 概要

担当員の処理する伝票を Fig.7 に示す OCR シートに転記し、適当な量になればカードあるいはフロッピーに変換し、大型機用、パソコン用のデータを作成する。実際には OCR (XONDEX-8000 日立造船情報システム) にデータシートを読ませるだけでよい。規則違反の文字については自動的にチェックされるが、転記ミスについては同時に出力されるリストをチェックしなければならない。前述の適当な量とはこのチェックが苦にならない程度をいう。この作業の結果、データが蓄積され、担当者が必要に応じて、集計・検索を行なうことができる。

Fig.7 Data format

例 : 通し番号 .. 22806
 NDC分類コード .. 121-59
 教員コード .. 130
 学部コード .. 1
 年度 .. 58
 月 .. 12
 金額 .. 18000
 洋・和書の分類 .. 1
 書籍名およびメモ .. サンキ` ヨウ.....

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31.....80



No	項目	桁数	備考
1	通し番号	5桁	
2	NDC分類コード	5桁	
3	教員コード	3桁	
4	学部コード	1桁	1:教養 2:経営 3:工学 4:短大
5	年度	2桁	
6	月	2桁	
7	金額	5桁	
8	洋・和書の分類	1桁	1:洋書 2:和書
9	書籍名およびメモ	45桁	

(2) 機能

処理機能として FORTRAN, BASIC とともに下記の機能を持っている。

- a. 追加, 訂正, 削除。
- b. 教員保管図書の検索……何年度から何年度までに購入した書籍のリストを出力する。
- c. NDC分類による検索……何年から何年までに購入した書籍から指定された分類に該当する書籍のリストを出力する。
- d. 集計……指定された年度, 学部の購入図書について, NDC大分類別, 和洋書別に冊数, 金額を月毎に集計し, 年度集計も行なう。

(3) 処理

FORTRAN で作成したシステムでは, 作成されたデータをディスクに保管し, 集計・検索の必要に応じて, バッチ処理, あるいはTSSによって出力を得る。BASIC で作成したシステムではフロッピーのコード変換を行なった後, 各年度毎のファイルをディスケットで保管する。事務処理が複数年度にまたがるときはハードディスク (10MB) に転送後に, 単年度の場合はディスケットのまま

Fig.8 A example of output

```

                222SENSEI

22802,0712 ,222,2,58,4 ,0,0,シ^
22803,0712 ,222,2,58,4 ,10000,0,シ^
23028,0712 ,222,2,58,5,0,0,シ^
:
:
23498,0712 ,222,2,58,11,..0,シ^
                                (書籍名)

洋書 **      0冊      和書 **      42冊      合計 **      42冊
金額 **      0円      金額 **144920円      金額 ** 144920円
    
```

58年度 合計						
分類	000	100	200	300	400	500
洋書	5 15420	35 367210	3 4640	192 1259337	10 73980	32 454660
和書	70 166780	107 298150	13 49680	198 678840	95 263710	188 933600
合計	75 182200	142 665360	16 54320	390 1938177	105 337690	220 1388260
分類	600	700	800	900	合計	
洋書	28 270380	2 12250	55 164060	60 376058	422 2997995	
和書	9 25200	33 111160	69 236670	49 189900	831 2953690	
合計	37 295580	35 123410	124 400730	109 565958	1253 5951685	

行なう。処理は対話形式でなされているので担当者はモニターを見ながら入力を行なう。このようにデータ作成の段階では共通性を持たせ、必要に応じて機械を使い分けることは事務効率をあげるだけでなく、従来より迅速かつ木目の細かいサービスをユーザに提供することが期待できる。今回は時間的制約もあり、Fig.7のデータシートに書籍名を記入しないままテストランを行なった。従って、集計業務（58年度）は実務として終了したが、検索に対するサービスは今後の課題となった。過去のデータを作成することも大変であるが、単年度の処理をパソコンで行ないながら逐次過去のデータをファイルにする方法が現実的と判断し、パソコン用のシステムについてはオペレーションマニュアルも作成し実務に耐えるようにしている。Fig.8に教員保管図書の検索（本来は書籍名も出力）と集計例を示す。なお、プログラムについてはリスト量が多いこと、当学独自のものであることから記載を控えた。なお、対象としたパソコンはPC9801、外部メモリーとして8"フロッピーディスク、5"ハードディスク（10MB）である。

3.3 マークカードによる試験の採点と集計

近年マークカードによる試験が広く行なわれ、その多くは選択肢から正答あるいは誤答を選ぶものである。このプログラムはマークカードによる試験を行なうことよりも、筆記試験との関連性を調べるために作成した。ところが理解度を知り、到達度の低い生徒に適切な指導ができるという思わぬメリットを得た。このため当初は大型機のみ処理を考えていたがクラス単位の処理をパソコンでも行ないたいという希望もあり、3.1と同様の使い方を検討している。以下にFORTRAN言語で開発したプログラムを紹介する。

(1) 概要

選択肢を選ぶ方法として1つの問題に対して数個の選択肢から回答する方法と、全問題に対して選択肢を選ぶ方法が考えられる。Fig.2の回答カードはその両方に使用できるように設計した。すなわち、1A, 1B……16A, 16Bは1問に対して9個の選択肢から選ぶ方法であり、後者は1, 2……16に対し、01~99の選択肢から選ぶ方法である。処理方法は全く同じであるが入力の手間を避けるために2つのプログラムとした。ここに示す例は前者の例である。

(2) 機能

採点終了後下記の出力を行なう。

- a. 学生番号, 序列, 得点一覧表
- b. 学生番号順の序列, 得点, 回答リスト
- c. 序列順に学生番号, 得点, 回答リスト
- d. 問題の正答率
- e. 得点度数分布表

(3) 処理

問題数, 各問題の配点, 各問題の正解, 回答カードの順に並べ, 処理を行なう。対象人数は一応700人としているので回答を貯えるだけで90KB程度の容量が必要となる。(2Byte単位の格納が可能であれば45KB)なお、回答マークカードの前につけるカードもマークカードを用い、1枚目の問題数は前2桁, 2枚目, 3枚目の配点, 正解カードは問題の回答欄に記入する。

プログラムとしては単純であるが、出力結果の使い方によっては教育効果の向上に役立つものと考えられる。Fig.9にプログラム, Fig.10に出力例の一部を示す。

Fig.9 Mark sense test program by FORTRAN

SOURCE STATEMENT

```

C
C*****
C SESEKI (モンタージュ 32ビット)
C*****
C 1982/05/24 (MON) *
      DIMENSION MD(700,3),NS(32),NK(32),NB(34)
      DIMENSION MN(700,34),NR(32)
      READ(5,500)M
500  FORMAT(I2)
      READ(5,501)(NS(I),I=1,M)
      READ(5,501)(NK(I),I=1,M)
501  FORMAT(10X,32I1)
      NO=0
      NG=0
      DO 15 I=1,32
15   NR(I)=0
      DO 10 I=1,700
      DO 10 J=1,3
      MD(I,J)=0
10   CONTINUE
100  READ(5,502,END=200)(NB(I),I=1,34)
502  FORMAT(I6,14,32I1)
      NO=NO+1
      MN(NO,1)=NB(1)
      MD(NO,1)=NB(1)
      NT=0
      DO 20 I=1,M
      IF(NB(I+2).NE.NK(I))GO TO 19
      NT=NT+NS(I)
      NR(I)=NR(I)+1
19   MN(NO,I+2)=NB(I+2)
20   CONTINUE
      NG=NG+NT
      MD(NO,2)=NT
      MN(NO,2)=NT
      GO TO 100
200  CALL JORETU(MD,NO)
      H=FLOAT(NG)/FLOAT(NO)
      CALL LIST(MD,NO)
      WRITE(6,603)H
603  FORMAT(1H0/1H ,8X,7HHEIKIN=,F6.2)
      CALL PLIST(NO,MN,NK,NR,M,MD)
      STOP
      END

      SUBROUTINE JORETU(MD,NO)
      DIMENSION MD(700,3)
      N=NO-1
      DO 10 I=1,N
      DO 20 J=I,NO
      IF(MD(I,1).LE.MD(J,1))GO TO 20
      DO 30 K=1,2
      IW=MD(I,K)
      MD(I,K)=MD(J,K)
      MD(J,K)=IW
30   CONTINUE
20   CONTINUE
10   CONTINUE
      NS=0
100  NJ=NS+1
      DO 40 I=1,NO
      IF(MD(I,3).NE.0)GO TO 40
      MAX=MD(I,2)
      NX=I
      GO TO 50

```

```

40 CONTINUE
   RETURN
50 DO 60 I=NX,NO
   IF(MD(I,3).NE.0)GO TO 60
   IF(MAX.GE.MD(I,2))GO TO 60
   MAX=MD(I,2)
60 CONTINUE
   DO 70 I=NX,NO
   IF(MAX.NE.MD(I,2))GO TO 70
   MD(I,3)=NJ
   NS=NS+1
70 CONTINUE
   GO TO 100
   END

```

```

SUBROUTINE LIST (MD,NO)
DIMENSION MD(700,3)
DO 10 I=1,NO,250
WRITE(6,600)
600 FORMAT(1H1,5(8H      NO,8H TOKUTEN,8H JORETSU))
   N=I+49
   DO 20 J=I,N
   IF(MOD(J,10)-1.EQ.0)WRITE(6,610)
610 FORMAT(1H )
   IF(J+200.LE.NO)GO TO 5
   IF(J+150.LE.NO)GO TO 4
   IF(J+100.LE.NO)GO TO 3
   IF(J+ 50.LE.NO)GO TO 2
   IF(J.LE.NO)GO TO 1
   GO TO 30
   5 WRITE(6,605)(MD(J,L),L=1,3),(MD(J+50,L),L=1,3),
   *(MD(J+100,L),L=1,3),(MD(J+150,L),L=1,3),(MD(J+200,L),L=1,3)
605 FORMAT(15I8)
   GO TO 20
   4 WRITE(6,604)(MD(J,L),L=1,3),(MD(J+50,L),L=1,3),
   *(MD(J+100,L),L=1,3),(MD(J+150,L),L=1,3)
604 FORMAT(12I8)
   GO TO 20
   3 WRITE(6,603)(MD(J,L),L=1,3),(MD(J+50,L),L=1,3),
   *(MD(J+100,L),L=1,3)
603 FORMAT(9I8)
   GO TO 20
   2 WRITE(6,602)(MD(J,L),L=1,3),(MD(J+50,L),L=1,3)
602 FORMAT(6I8)
   GO TO 20
   1 WRITE(6,601)(MD(J,L),L=1,3)
601 FORMAT(3I8)
20 CONTINUE
10 CONTINUE
30 RETURN
   END

```

```

SUBROUTINE PLIST(NO,MN,NK,NR,M,MD)
DIMENSION MN(700,34),NK(32),NR(32),NN(32),MD(700,3)
DO 60 I=1,M
60 NN(I)=I
   N=NO-1
   DO 110 I=1,N
   DO 120 J=I,NO
   IF(MN(I,1).LE.MN(J,1)) GO TO 120
   DO 130 K=1,34
   L=MN(I,K)
   MN(I,K)=MN(J,K)
   MN(J,K)=L
130 CONTINUE
120 CONTINUE
110 CONTINUE
   DO 150 K=1,NO
   IF(MOD(K,50)-1.NE.0)GO TO 160
   WRITE(6,701) (NN(I),I=1,M)
701 FORMAT(1H1,14X,32I3)
   WRITE(6,702) (NK(I),I=1,M)
702 FORMAT(1H0,10X,4H1カ1,32I3)
160 IF(MOD(K,10)-1.EQ.0)WRITE(6,613)
   WRITE(6,703)(MD(K,J),J=1,3),(MN(K,J),J=3,34)
150 CONTINUE
703 FORMAT(1H ,16,2I4,32I3)
   WRITE(6,600) NO
600 FORMAT(1H1//1H0,'*****カイツ*****イトウズ',I6)
   WRITE(6,601)
601 FORMAT(1H0,3X,2HNO,3X,2HOK,4X,1H%)
   DO 10 I=1,M
   R=FLOAT(NR(I))/FLOAT(NO)*100.0
   WRITE(6,605) I,NR(I),R
   10 CONTINUE
605 FORMAT(1H ,2I5,F8.2)
   N=NO-1
   DO 20 I=1,N
   DO 30 J=I,NO
   IF(MN(I,2).GE.MN(J,2))GO TO 30
   DO 40 K=1,34
   L=MN(I,K)
   MN(I,K)=MN(J,K)
   MN(J,K)=L
40 CONTINUE
30 CONTINUE
20 CONTINUE
   DO 100 I=1,NO
   IF(MOD(I,50)-1.NE.0)GO TO 90
   WRITE(6,610) (NN(J),J=1,M)
   WRITE(6,611) (NK(J),J=1,M)

```

```

610 FORMAT (1H1,10X,32I3)
611 FORMAT (1H0,6X,4H1カ1,32I3)
90 IF(MOD(I,10)-1.EQ.0)WRITE(6,613)
613 FORMAT(1H )
WRITE(6,615)(MN(I,J),J=1,34)
615 FORMAT(1H ,I6,I4,32I3)
100 CONTINUE
CALL DOS(MD,NO)
RETURN
END

```

```

SUBROUTINE DOS(MD,NO)
DIMENSION MD(700,3),KD(10),NP(5,100),LD(10)
DATA IT/1H./,IA/1H*/,IB/1H /
DO 10 I=1,10
10 KD(I)=0
DO 20 J=1,NO
N=MD(J,2)/10
IF(N.EQ.10)N=9
KD(N+1)=KD(N+1)+1
20 CONTINUE
DO 30 I=1,10
30 LD(I)=I*10
WRITE(6,600)(LD(I),I=1,10)
600 FORMAT(1H1,20X,10I10)
DO 50 I=1,10
DO 60 J=1,5
DO 70 K=1,100
NP(J,K)=IB
IF(MOD(K,5).EQ.0)NP(J,K)=IT
70 CONTINUE
60 CONTINUE
M=KD(I)
IF(M.LE.500)GO TO 76
75 STOP
76 DO 80 J=1,M
IF(J.LE.100)NP(1,J)=IA
IF(J.GT.100.AND.J.LE.200)NP(2,J-100)=IA
IF(J.GT.200.AND.J.LE.300)NP(3,J-200)=IA
IF(J.GT.300.AND.J.LE.400)NP(4,J-300)=IA
IF(J.GT.400.AND.J.LE.500)NP(5,J-400)=IA
80 CONTINUE
KK=I*10-1
DO 90 J=1,5
IF(J.EQ.3)WRITE(6,601)KK
WRITE(6,602)(NP(J,K),K=1,100)
90 CONTINUE
WRITE(6,603)
50 CONTINUE
601 FORMAT(1H+,I3,I7)
602 FORMAT(1H ,20X,100A1)
603 FORMAT(1H )
RETURN
END

```


Fig.10 A example of output

NO TOKUTEN JORETSU			NO TOKUTEN JORETSU			NO TOKUTEN JORETSU		
210101	78	313	210214	60	509	210331	61	498
210102	94	106	210215	80	286	210333	84	244
210103	54	537	210216	100	1	210334	60	509
210104	47	556	210217	91	162	210335	86	220
210105	96	43	210218	92	124	210336	69	420
210106	71	391	210219	12	582	210337	100	1
210107	61	498	210220	87	210	210338	62	490
210108	65	462	210221	59	511	210339	61	498
210109	86	220	210231	96	43	210340	70	
210110	90	168	210232	92	124	210341		

210111 98
210112

			1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
	セイカイ		4	8	3	7	6	2	5	3	2	1	4	5	1	2	3	5	1	3	7	
210101	78	313	4	8	3	7	6	1	5	3	2	2	4	5	1	2	3	5	3	8	7	
210102	94	106	4	8	3	7	6	2	5	3	2	1	4	5	1	2	3	5	1	3	7	
210103	54	537	4	8	3	7	6	2	5	1	2	3	4	5	2	1	3	2	5	3	7	
210104	47	556	4	8	3	7	2	6	5	5	2	3	4	5	1	2	3	3	5	8	6	
210105	96	43	4	8	3	7	6	2	5	3	3	1	4	5	1	2	3	5	1	3	7	
210106	71	391	4	8	1	7	6	3	5	5	1	3	4	5	1	2	3	5	1	3	7	4
210107	61	498	4	8	3	7	6	9	5	1	2	3	4	5	1	2	3	1	8	2	7	6
210108	65	462	4	8	3	7	2	1	5	2	1	2	4	5	1	2	3	5	1	3	7	6
210109	86	220	4	8	3	7	6	2	5	3	2	3	4	5	1	2	3	1	5	3	7	4
210110	90	168	4	8	3	7	6	1	5	1	2	3	4	5	1	2	3	5	1	3	7	4
210111	98	36	4	8	3	7	6	1	5	3	2	1	4	5	1	2	3	5	1	3	7	
210112	92	124	4	8	3	7	6	2	5	3	2	3	4	5	1	2	3	5	1	3	7	
210113	73	359	4	8	3	7	6	2	5	5	2	1	4	5	1	2	3	7	5	3	8	

.

*****セイヤイリツ***セイトソウ 620

	NO	OK	%
59	1	611	98.55
	2	598	96.45
	3	600	96.77
69	4	586	94.52
	5	588	94.84
	6	435	70.16
	7	553	89.19
	8	453	73.06
	9	522	84.19
79	10	514	82.90
	11	559	90.16
	12	537	
	13		

89 *****

99 *****

.

3. 4 中学用 5 教科試験集計システム

中学校における教員の仕事量は想像以上に多い。最終学年ともなれば尚更である。特にクラス数の多い学校では事務量も級数的に増加するようで、生徒が倍になれば教員を倍にすればよいという単純思考では教育問題は解決しないように思われる。このような、いわゆるマンモス校でもパソコンで成績集計ができないか、その延長線上でこれらのデータを基にした種々の処理ができればという発想で本システムを開発した。また、1 学年700人程度の処理を P C 8001 程度の機種で行ない、しかも操作、入力方法を容易にすることを目的としてシステム作成を行なった。

以下にその説明を行なう。

(1) 概要

入力の方法として、あらかじめ生徒名簿を作成しておけば、任意のクラスに任意の科目点数が入力できる。入力が終了すると 5 科目合計 (国, 数, 社, 理, 英), 全学年序列, 偏差値と, 参考として 3 科目合計 (国, 数, 英) を出力する。またクラス別の各教科の平均点, 各科目の度数分布表も資料として出力できる。これらのプログラムの機能については後述するが, これらのデータプログラムはディスクに格納し小さい容量でも処理できるように工夫した。また並べ替えの方法も quick sort を採用し処理時間も 1/10 程度に短縮した。

(2) 機能

本システムは, 国語, 社会, 数学, 理科, 英語の 5 科目試験の結果を集計するシステムで, 下記のプログラムから構成されている。

1. 原本 (名簿) 作成プログラム

名簿をクラス別に作成し, ドライブ 1 のディスクに格納する。

- a) プログラム名 “genpon”
- b) 最大クラス数 14 クラス
- c) 1 クラス最大人数 49 名

2. 作業用名簿作成プログラム

原本の名簿をドライブ 2 のディスクに移す。

- b) プログラム名 “name”

3. 成績入力プログラム

クラス別, 科目別に成績を入力する。

- a) プログラム名 “seiseki”

4. チェックプログラム

入力された氏名, 成績をチェックするために出力するプログラムである。

また集計終了後に, このプログラムを使用すると, 処理結果も出力できる。

- a) プログラム名 “check”

5. 作成された名簿, あるいは成績を訂正したり, 名簿の追加, 削除を行なう。このプログラムは作業用ファイルの訂正, 追加, 削除だけでなく, 原本の訂正, 追加, 削除も行なえる。

- a) プログラム名 “teisei”

6. 成績クリアプログラム

指示したクラスの成績をすべて clear するプログラムで “check” や後述の “qsum” の実行中にエラーが出た時, 全成績を clear させる

- a) プログラム名 “clear”

7. 集計プログラム

成績入力が終了すると, 3 科目合計, 5 科目合計, 偏差値, 学年序列を計算し出力する。

- a) プログラム名 “qsum”
8. 資料作成プログラム
各クラスと全体の科目別（3計・5計も含む）平均点および科目別の度数表を出力する。

a) プログラム名 “shiryo”

(3) オペレーション

前述のプログラムは、メニュー方式で構成しているため、プログラムの呼び出しのために MOUNT, LOAD を、逐一しなくてよい。

呼び出しの方法は、下記の通りである。

1. ドライブ1（以下d1）に、システムプログラムを入れる。
2. ドライブ2（以下d2）に、データ用ディスクセットを挿入する。
3. MOUNT1〔R/T〕
4. LOAD“M”〔R/T〕
5. 呼び出したいプログラム番号〔R/T〕 auto-run になっているため RUN の指示がなくとも実行に入る。
6. プログラムが終了し、再度実行したいときは〔f5〕（run）,
他のプログラムを呼び出したいときは〔f1〕を押すと、メニューが表示されるので、No.5からの繰返しを、行なう。
7. 終了するときも、〔f1〕を押し、メニューの9番（*オワリ*）を入力する。

以下にそのプログラムと出力例を示すが、このプログラムはいろいろ使い道があるようで、進路指導用、担当科目の総合評価等のプログラムへの改良もされている。我々サイドでも学生の実験、実習の評価システムへの改良も検討している。また次の問題としてこれらのデータを大型機に移し、個々の生徒に対する目目細かい指導資料の作成を検討している。Fig.11 にプログラム、Fig.12 に出力例を示す。

Fig.11 Totalization program of examination by BASIC

```

10 '
20 ' |   メニュー   (1983年 9月) |
30 '
35 REMOVE
40 PRINT CHR$(12) : DIM F$(8)
50 PRINT TAB(10);"===== "
60 PRINT TAB(10);"=                ="
70 PRINT TAB(10);"=          メニュー          ="
80 PRINT TAB(10);"=                ="
90 PRINT TAB(10);"===== "
100 PRINT
110 PRINT TAB(12);"1. genpon      " : F$(1)="genpon"
120 PRINT TAB(12);"2. name       " : F$(2)="name"
130 PRINT TAB(12);"3. seisek    " : F$(3)="seisek"
140 PRINT TAB(12);"4. teisei    " : F$(4)="teisei"
150 PRINT TAB(12);"5. qsum      " : F$(5)="qsum"
160 PRINT TAB(12);"6. shiryo   " : F$(6)="shiryo"
170 PRINT TAB(12);"7. check    " : F$(7)="check"
180 PRINT TAB(12);"8. clear    " : F$(8)="clear"
185 PRINT TAB(12);"9. * オワリ *  "
190 PRINT TAB(30);: INPUT "No.";Z
195 IF Z=9 THEN 240
200 IF Z<0 OR Z>8 THEN 190

```

```

210 '
220 KEY 1,"mount1:run"+CHR$(34)+"M"+CHR$(34)+CHR$(13)
230 MOUNT 1 : RUN F$(Z)
235 '
240 CLOSE : REMOVE
250 PRINT CHR$(12)
260 LOCATE 20,10 : PRINT "オツカレサマ !! "
270 END

```

```

100 PRINT CHR$(12)
110 PRINT " genpon "
120 PRINT " | genpon | "
130 PRINT " genpon "
140 DEFINT Q
150 '
160 GOSUB 380
170 '
180 DIM QN(50),NM$(50),Q(50,9)
190 PRINT : INPUT "ナンクミ テゝスカ " ;K
200 PRINT : PRINT K;"クミ ノ ニンスウ " ;:INPUT NO
210 A$=CN$(K) : PRINT
220 FOR I=1 TO NO
230 INPUT "No. " ;QN(I) : QN(I)=K*100+QN(I)
240 NEXT I
250 BEEP
260 '
270 FOR I=1 TO NO
280 PRINT QN(I);"name " ;: INPUT NM$(I)
290 NEXT I
300 BEEP
310 '
320 GOSUB 500
330 '
340 INPUT "ホカ ノ クラスモ ニユリヨク シマスカ ? 1.yes 2.no " ;Y
350 IF Y=1 THEN 190
360 END
370 '
380 ' =====
390 ' == SUB 1000 ==
400 ' =====
410 DIM CN$(14)
420 FOR I=1 TO 14
430 READ CN$(I)
440 NEXT I
450 '
460 DATA t01,t02,t03,t04,t05,t06,t07
470 DATA t08,t09,t10,t11,t12,t13,t14
480 RETURN
490 '
500 ' =====
510 ' == SUB 2000 ==
520 ' =====
530 REMOVE : MOUNT 1

```

```

540 OPEN A$ FOR OUTPUT AS #1
550 PRINT #1,NO
560 FOR II=1 TO NO
570 PRINT #1,QN(II),NM$(II)
580 NEXT II
590 '
600 FOR II=1 TO NO
610 FOR IJ=1 TO 9
620 PRINT #1,Q(II,IJ)
630 NEXT IJ,II
640 CLOSE #1 : REMOVE
650 RETURN

```

```

100 '
110 ' | name store from D1 to D2 (1983年 9月) |
120 '
130 DEFINT Q : PRINT CHR$(12)
132 PRINT " |
134 PRINT " | name |
136 PRINT " |
138 PRINT
140 DIM QN(50),NM$(50),Q(50,9)
150 '
160 GOSUB 1000
170 '
180 INPUT "ナンクミ ヲ ウツシマスカ":K
185 PRINT
190 A$=CN$(K)
200 '
210 GOSUB 3000
220 '
230 GOSUB 2000
240 '
250 INPUT "ホカノ クラスモウツシマスカ 1=no, 2=yes ";Y
255 PRINT
260 IF Y=2 THEN 180
270 END
280 '
1000 ' =====
1010 ' == SUB 1000 ==
1020 ' =====
1030 DIM CN$(14)
1040 FOR I=1 TO 14
1050 READ CN$(I)
1060 NEXT I
1065 '
1070 DATA t01,t02,t03,t04,t05,t06,t07
1080 DATA t08,t09,t10,t11,t12,t13,t14
1090 RETURN
1100 '
2000 ' =====
2010 ' == SUB 2000 ==
2020 ' =====
2030 REMOVE : MOUNT 2
2040 OPEN"2:"+A$ FOR OUTPUT AS #1

```

```

2050 PRINT #1,NO
2060 FOR II=1 TO NO
2070   PRINT #1,QN(II),NM$(II)
2080 NEXT II
2085 '
2090 FOR II=1 TO NO
2100   FOR IJ=1 TO 9
2110     PRINT #1,Q(II,IJ)
2120   NEXT IJ,II
2130 CLOSE #1 : REMOVE
2140 RETURN
2150 '
3000 ' =====
3010 ' ==      SUB 3000      ==
3020 ' =====
3030 REMOVE : MOUNT 1
3040 OPEN A$ FOR INPUT AS #1
3050 INPUT #1,NO
3060 FOR II=1 TO NO
3070   INPUT #1,QN(II),NM$(II)
3080 NEXT II
3085 '
3090 FOR II=1 TO NO
3100   FOR IJ=1 TO 9
3110     INPUT #1,Q(II,IJ)
3120   NEXT IJ,II
3130 CLOSE #1 : REMOVE
3140 RETURN

```

```

100 '
110 ' | check (1983年 9月) |
120 '
121 PRINT CHR$(12)
122 PRINT " | "
123 PRINT " | check | "
124 PRINT " | "
130 DEFINT Q : LPRINT CHR$(12)
140 '
150 GOSUB 1000
160 '
170 DIM QN(50),NM$(50),Q(50,9)
180 PRINT : INPUT "print スル クラス";K
190 A$=CN$(K)
200 '
210 GOSUB 3000
220 '
230 LPRINT K;"クミ";NO;"ニン" : LPRINT
232 '
235 GOSUB 7000
237 '
240 FOR I=1 TO NO
250   LPRINTUSING " #### &           &":QN(I);NM$(I):
260   FOR J=1 TO 9
270     LPRINTUSING " ###":Q(I,J):
280   NEXT J
290   LPRINT

```

```

300 NEXT I
310 END
320 '
1000 ' =====
1010 ' ==      SUB 1000      ==
1020 ' =====
1030 DIM CN$(14)
1040 FOR I=1 TO 14
1050   READ CN$(I)
1060 NEXT I
1070 '
1080 DATA t01,t02,t03,t04,t05,t06,t07
1085 DATA t08,t09,t10,t11,t12,t13,t14
1090 RETURN
1100 '
3000 ' =====
3010 ' ==      SUB 3000      ==
3020 ' =====
3030 REMOVE : MOUNT 2
3040 OPEN "2:"+A$ FOR INPUT AS #1
3050   INPUT #1,NO
3060   FOR II=1 TO NO
3070     INPUT #1,QN(II),NM$(II)
3080   NEXT II
3090 '
3100   FOR II=1 TO NO
3110     FOR IJ=1 TO 9
3120       INPUT #1,Q(II,IJ)
3130     NEXT IJ,II
3140 CLOSE #1 : REMOVE
3150 RETURN

```

```

100 '
110 ' | teisei      (1983年 9月) |
120 '
130 DEFINT Q : PRINT CHR$(12)
132 PRINT " | teisei |"
134 PRINT " | teisei |"
136 PRINT " | teisei |"
138 PRINT
140 '
150 GOSUB 1000
160 '
170 DIM QN(50),NM$(50),Q(50,9)
180 PRINT : INPUT "テイセイスル クラス ";K
190 A$=CN$(K)
200 '
210 GOSUB 3000
220 '
225 PRINT
230 INPUT "テイセイハ 1.メイホ` 2.セイセキ" " :T
240 IF T=2 THEN 640
250 '
260 ' メイホ` テイセイ
270 '

```

```

275 PRINT
280 INPUT "1.テイセイ 2.ツイカ 3.サクシヨ      ";TA
290 IF TA<>1 THEN 380
295 PRINT
300 INPUT "テイセイスル セイトハ`ンコ`ウ      No.=";QA
310 FOR I=1 TO NO
320   IF QN(I)=QA THEN 350
330 NEXT I
340 PRINT : PRINT "セイトハ`ンコ`ウ ナシ !!" : GOTO 300
350 PRINT : INPUT "New No. Name ";QN(I).NM$(I)
360 GOTO 730
370 '
380 IF TA<>2 THEN 510
390 '
400 ' メイホ` ツイカ
410 '
420 PRINT : INPUT "ナンハ`ンメ ニ ツイカシマスカ ";NK
430 NO=NO+1
435 IF NO=NK THEN 480
440 FOR I=NO TO NK STEP -1
450   IF I=1 THEN 480
460   QN(I)=QN(I-1) : NM$(I)=NM$(I-1)
470 NEXT I
480 PRINT : INPUT "New No. Name ";QN(NK).NM$(NK)
490 GOTO 730
500 '
510 ' メイホ` サクシヨ
520 '
530 PRINT : INPUT "サクシ`ヨスル セイトハ`ンコ`ウ ";QA
540 M=NO : NO=NO-1
550 IF QN(M)=QA THEN 730
560 FOR I=1 TO M
570   IF QN(I)=QA THEN 590
580 NEXT I
590 FOR J=I+1 TO M
600   QN(J-1)=QN(J) : NM$(J-1)=NM$(J)
610 NEXT J
620 GOTO 730
630 '
640 ' セイセキ テイセイ
650 '
660 PRINT : INPUT "セイト ハ`ンコ`ウ ";QA
670 FOR I=1 TO NO
680   IF QN(I)=QA THEN 700
690 NEXT I
700 PRINT : INPUT "テイセイスル カモク 1.コク 2.シャ 3.スウ 4.リカ 5.エイ ";J
710 PRINT : INPUT "テイセイテン ";Q(I,J)
720 '
730 PRINT : INPUT "ホカノ テイセイ 0.ナシ 1.アリ ";Y
740 IF Y=1 THEN 225
750 '
760 GOSUB 2000
770 '
780 PRINT : INPUT "ケンホ`ンモ コウシンシマスカ 1.yes 2.no ";T
790 IF T=2 THEN END
800 '
810 REMOVE : MOUNT 1
820 OPEN A$ FOR OUTPUT AS #1

```



```

830 PRINT #1,NO
840 FOR I=1 TO NO
850 PRINT #1,QN(I),NM$(I)
860 NEXT I
870 '
880 FOR I=1 TO NO
890   FOR J=1 TO 9
900     PRINT #1,Q(I,J)
910   NEXT J,I
920 CLOSE #1 : REMOVE
930 END
940 '
1000 ' =====
1010 ' ==      SUB  1000      ==
1020 ' =====
1030 DIM CN$(14)
1040 FOR I=1 TO 14
1050   READ CN$(I)
1060 NEXT I
1070 '
1080 DATA t01,t02,t03,t04,t05,t06,t07
1090 DATA t08,t09,t10,t11,t12,t13,t14
1100 RETURN
1110 '
2000 ' =====
2010 ' ==      SUB  2000      ==
2020 ' =====
2030 REMOVE : MOUNT 2
2040 OPEN "2:"+A$ FOR OUTPUT AS #1
2050 PRINT #1,NO
2060 FOR II=1 TO NO
2070   PRINT #1,QN(II),NM$(II)
2080 NEXT II
2090 '
2100 FOR II=1 TO NO
2110   FOR IJ=1 TO 9
2120     PRINT #1,Q(II,IJ)
2140 CLOSE #1 : REMOVE
2150 RETURN
2160 '
3000 ' =====
3010 ' ==      SUB  3000      ==
3020 ' =====
3030 REMOVE : MOUNT 2
3040 OPEN "2:"+A$ FOR INPUT AS #1
3050 INPUT #1,NO
3060 FOR II=1 TO NO
3070   INPUT #1,QN(II),NM$(II)
3080 NEXT II
3090 '
3100 FOR II=1 TO NO
3110   FOR IJ=1 TO 9
3120     INPUT #1,Q(II,IJ)
3130   NEXT IJ,II
3140 CLOSE #1 : REMOVE
3150 RETURN

```

```

100 '
110 '   qsum   (1983年 9月)
120 '
130 PRINT CHR$(12)
140 PRINT "
150 PRINT "   qsum
160 PRINT "
170 '
180 DEFINT Q
190 PRINT :PRINT
200 INPUT "クラス ソウスウ";KK
210 '
220 DIM QN(50),NM$(50),Q(50,9)
230 DIM QB(700),QG(700),QJ(700)
240 '
250 GOSUB 1000
260 '
270 N=0 : GT=0 : GG=0
280 FOR I=1 TO KK
290   A$=CN$(I)
300   PRINT : PRINT I ; " クミ ヨミコミチュウ"
310   GOSUB 3000
320 '
330   N=N+NO : LS=N-NO
340   FOR J=1 TO NO
350     Q(J,6)=Q(J,1)+Q(J,3)+Q(J,5)
360     Q(J,7)=Q(J,6)+Q(J,2)+Q(J,4)
370 '
380     QB(LS+J)=QN(J) : QG(LS+J)=Q(J,7) : GT=GT+Q(J,7)
390     GG=GG+Q(J,7)^2
400   NEXT J
410   PRINT : PRINT I ; " クミ カキコミチュウ"
420   GOSUB 2000
430 '
440 NEXT I
450 H=GT/N
460 S=(GG/N-H^2)^.5
470 PRINT : PRINT "ナラヘ`カエテ イマス"
480 GOSUB 5000
490 '
500 FOR I=1 TO N
510   QJ(I)=I
520 NEXT I
530 '
540 FOR I=2 TO N
550   IF QG(I)=QG(I-1) THEN QJ(I)=QJ(I-1)
560 NEXT I
570 PRINT : PRINT "** ウチタ`シ ト ナラヘ`カエ ヲ シテイマス"
580 GOSUB 4000 : GOSUB 5220
590 '
600 M=0
610 FOR I=1 TO KK
620   A$=CN$(I)
630   PRINT : PRINT I ; " クミ ノ イチラン ウチタ`シ"
640   GOSUB 3000
650 '
660   GOSUB 7000
670 '
680   M=M+NO : LS=M-NO

```

```

690   FOR J=1 TO NO
700     Q(J,9)=QJ(LS+J)
710     Q(J,8)=50+INT(Q(J,7)-H)/S*10
720     LPRINTUSING " #### &          &";QN(J);NM$(J);
730   '
740     FOR L=1 TO 9
750       LPRINTUSING " ###";Q(J,L);
760     NEXT L
770     LPRINT
780   NEXT J
790   PRINT : PRINT I ;" クミ DATA ヲ Disk へ カキコミチュウ"
800   GOSUB 2000
810   '
820 NEXT I
830 END
840 '
1000 ' =====
1010 ' ==      SUB 1000      ==
1020 ' =====
1030 DIM CN$(14) : RESTORE 1080
1040 FOR I=1 TO 14
1050   READ CN$(I)
1060 NEXT I
1070 '
1080 DATA t01,t02,t03,t04,t05,t06,t07
1090 DATA t08,t09,t10,t11,t12,t13,t14
1100 RETURN
1110 '
2000 ' =====
2010 ' ==      SUB 2000      ==
2020 ' =====
2030 REMOVE : MOUNT 2
2040 OPEN "2:"+A$ FOR OUTPUT AS #1
2050   PRINT #1,NO
2060   FOR II=1 TO NO
2070     PRINT #1,QN(II),NM$(II)
2080   NEXT II
2090 '
2100   FOR II=1 TO NO
2110     FOR IJ=1 TO 9
2120       PRINT #1,Q(II,IJ)
2130     NEXT IJ,II
2140   CLOSE #1 : REMOVE
2150   RETURN
2160 '
3000 ' =====
3010 ' ==      SUB 3000      ==
3020 ' =====
3030 REMOVE : MOUNT 2
3040 OPEN "2:"+A$ FOR INPUT AS #1
3050   INPUT #1,NO
3060   FOR II=1 TO NO
3070     INPUT #1,QN(II),NM$(II)
3080   NEXT II
3090 '
3100   FOR II=1 TO NO
3110     FOR IJ=1 TO 9
3120       INPUT #1,Q(II,IJ)
3130     NEXT IJ,II

```

```

3140 CLOSE #1 : REMOVE
3150 RETURN
3160 '
4000 ' =====
4010 ' ==      SUB 4000      ==      (ｼﾞｮｲﾝﾄﾞ ﾘｽﾄ)
4020 ' =====
4030 '
4040 FOR I=1 TO N STEP 5
4050 IF ((I-1) MOD 250)<>0 THEN 4100
4060 LPRINT CHR$(12) : LPRINT " ";
4070 FOR J=1 TO 5
4080 LPRINT USING " # ";J;
4090 NEXT J : LPRINT
4100 IF ((I-1)MOD 50)=0 THEN LPRINT
4110 LPRINT USING "###- ";I;
4120 FOR J=I TO I+4
4130 IF QB(J)=0 THEN 4150
4140 LPRINT USING " #### ### ### ";QB(J),QG(J),QJ(J);
4150 NEXT J : LPRINT
4160 NEXT I
4170 RETURN
4180 '
5000 ' =====
5010 ' ==      SUB 5000      ==      (quick sort1)
5020 ' =====
5030 DIM K0(20),K1(20)
5040 K=1:K0(1)=1:K1(1)=N
5050 L=K0(K):R=K1(K):K=K-1
5060 I=L:J=R:W=QG((L+R)/2)
5070 IF QG(I)>W THEN I=I+1:GOTO 5070
5080 IF W>QG(J) THEN J=J-1:GOTO 5080
5090 IF I<=J THEN SWAP QG(I),QG(J):SWAP QB(I),QB(J):I=I+1:J=J-1
5100 IF I<=J THEN 5070
5110 IF J-L>=R-I THEN 5150
5120 IF I<R THEN K=K+1:K0(K)=I:K1(K)=R
5130 R=J
5140 GOTO 5170
5150 IF L<J THEN K=K+1:K0(K)=L:K1(K)=J
5160 L=I
5170 IF L<R THEN 5060
5180 IF K<>0 THEN 5050
5190 ERASE K0,K1
5200 RETURN
5210 '
5220 ' =====
5230 ' ==      SUB 6000      ==      (quick sort 2)
5240 ' =====
5250 DIM K0(20),K1(20)
5260 K=1:K0(1)=1:K1(1)=N
5270 L=K0(K):R=K1(K):K=K-1
5280 I=L:J=R:W=QB((L+R)/2)
5290 IF QB(I)<W THEN I=I+1:GOTO 5290
5300 IF W<QB(J) THEN J=J-1:GOTO 5300
5310 IF I<=J THEN SWAP QB(I),QB(J):SWAP QJ(I),QJ(J):I=I+1:J=J-1
5320 IF I<=J THEN 5290
5330 IF J-L>=R-I THEN 5370
5340 IF I<R THEN K=K+1:K0(K)=I:K1(K)=R
5350 R=J
5360 GOTO 5390

```

```

5370 IF L<J THEN K=K+1:K0(K)=L:K1(K)=J
5380 L=I
5390 IF L<R THEN 5280
5400 IF K<>0 THEN 5270
5410 ERASE K0,K1
5420 RETURN
5430 '
7000 ' =====
7010 ' ==      SUB 7000      ==   (カモクメイ ウチタシ)
7020 ' =====
7030 '
7040 LPRINT CHR$(12) : RESTORE 7310
7050 LPRINT TAB(15);
7060 FOR JM=1 TO 5
7070 LPRINT "  ";
7080 GOSUB 7210
7090 NEXT JM
7100 '
7110 LPRINT "  3";
7120 GOSUB 7210
7130 LPRINT "  5";
7140 GOSUB 7210
7150 FOR JM=1 TO 2
7160 LPRINT "  ";
7170 GOSUB 7210
7180 NEXT JM
7190 LPRINT : LPRINT
7200 RETURN
7210 LPRINT CHR$(&H1B);"I";"0015";
7220 FOR IM=1 TO 30
7230  READ D$ : C=VAL("&h"+D$) : GOSUB 7270
7240 NEXT IM
7250 RETURN
7260 '
7270 '**** PUT CHARACTORE for PC-8801,8001****
7280 A=INP(&H40) AND 1 : IF A<>0 THEN 7280
7290 OUT &H10,C : OUT &H40,0 : OUT &H40,1 : RETURN
7300 '
7310 ' コク
7320 DATA 00,00,fe,ff,02,80,02,90,12,91,12,91,12,91,f2,9f
7330 DATA 12,91,12,91,12,95,12,99,02,90,fe,ff,00,00
7340 ' シャ
7350 DATA 20,08,22,04,22,02,24,ff,24,03,a8,04,68,08,20,40
7360 DATA 00,40,40,40,40,40,ff,7f,40,40,40,40,00,40
7370 ' スウ
7380 DATA 49,84,2a,5c,08,27,ff,64,08,94,2a,0c,49,04,20,80
7390 DATA 1f,40,08,30,c8,0c,08,03,c8,08,28,30,08,c0
7400 ' リ
7410 DATA 00,00,42,20,42,20,fe,3f,42,10,42,08,42,08,00,20
7420 DATA 3e,21,2a,21,2a,21,fe,3f,2a,21,2a,21,3e,20
7430 ' エイ
7440 DATA 00,82,04,42,04,22,ef,23,24,12,24,12,24,0a,e4,07
7450 DATA 24,0a,24,12,2c,12,e7,23,04,22,04,42,00,82
7460 ' ケイ
7470 DATA 04,00,24,f9,24,89,27,89,24,f9,04,00,44,00,40,00
7480 DATA 40,00,40,00,ff,ff,40,00,40,00,40,00,40,00
7490 ' ケイ
7500 DATA 04,00,24,f9,24,89,27,89,24,f9,04,00,44,00,40,00
7510 DATA 40,00,40,00,ff,ff,40,00,40,00,40,00,40,00

```

```

7520 ' ^
7530 DATA 00,03,c0,00,f8,ff,07,04,80,03,78,ff,49,09,49,09
7540 DATA 49,ff,49,09,49,09,49,ff,49,09,49,49,70,ff
7550 ' ショ
7560 DATA 00,c0,00,30,fc,0f,04,00,14,02,14,02,14,22,57,42
7570 DATA 94,fe,54,03,34,02,14,0a,04,06,00,02,00,00
7580 END

```

```

10 '
20 ' 

|        |            |
|--------|------------|
| shiryo | (1983年 9月) |
|--------|------------|


30 '
31 PRINT CHR$(12)
32 PRINT "
33 PRINT "| shiryo |"
34 PRINT "
40 DEFINT Q
50 DIM QN(50),NM$(50),Q(50,9)
60 DIM TS(14,7),TT(7),QK(14),QG(7,10)
70 '
80 GOSUB 1000
90 '
100 PRINT : INPUT "ツウクラス スウ ";KK
110 N=0
120 FOR II=1 TO KK
130 A$=CN$(II)
140 '
150 PRINT II;" クミ ショリチュウ !" : GOSUB 3000
160 '
170 N=N+NO : QK(II)=NO
180 FOR I=1 TO 7
190 FOR J=1 TO NO
200 TS(II,I)=TS(II,I)+Q(J,I)
210 TT(I)=TT(I)+Q(J,I)
220 IF I>5 THEN 250
230 IF Q(J,I)=100 THEN Q(J,I)=Q(J,I)-1
240 X=INT(Q(J,I)/10)+1 : GOTO 300
250 IF I=7 THEN 280
260 IF Q(J,I)=300 THEN Q(J,I)=Q(J,I)-1
270 X=INT(Q(J,I)/30)+1 : GOTO 300
280 IF Q(J,I)=500 THEN Q(J,I)=Q(J,I)-1
290 X=INT(Q(J,I)/50)+1
300 QG(I,X)=QG(I,X)+1
310 NEXT J,I
320 '
330 FOR I=1 TO 7
340 TS(II,I)=TS(II,I)/NO
350 NEXT I
360 NEXT II
370 '
380 FOR I=1 TO 7
390 TT(I)=TT(I)/N
400 NEXT I
410 '
420 DIM QQ$(7) : RESTORE 470
430 FOR I=1 TO 7
440 READ QQ$(I)
450 NEXT I

```

```

460 '
470 DATA コクコ~,シャカイ,スウカ`ク,リカ,エイコ~,3ケイ,5ケイ
480 '
490 INPUT "コヒ° - マイスウ ";CP
500 FOR PP=1 TO CP
510 LPRINT CHR$(12)
520 LPRINT "ソウスウ ";N;" ニン" : LPRINT : LPRINT
530 LPRINT "カモクハ`ツ ハイキン" : LPRINT
540 LPRINT " クミ";TAB(6);"コクコ`";TAB(12);"シャカイ";TAB(18);"スウカ`ク";
550 LPRINT TAB(24);"リカ";TAB(30);"エイコ`";TAB(35);"3ケイ";TAB(41);"5ケイ"
560 FOR I=1 TO KK
570 LPRINT USING " ###";I;
580 FOR J=1 TO 7
590 LPRINT USING " ###.#";TS(I,J);
600 NEXT J
610 LPRINT
620 NEXT I
630 '
640 LPRINT : LPRINT " TT";
650 FOR I=1 TO 7
660 LPRINT USING " ###.#";TT(I);
670 NEXT I
680 LPRINT : LPRINT : LPRINT : LPRINT : LPRINT
690 '
700 LPRINT "フ`ンフ" : LPRINT
710 LPRINT SPACE$(10);
720 FOR I=9 TO 89 STEP 10
730 LPRINT USING " -##";I;
740 NEXT I
750 LPRINT "-100 (マテ`)"
760 LPRINT
770 FOR I=1 TO 7
780 LPRINT USING "& &";QQ$(I);" --- ";
790 FOR J=1 TO 10
800 LPRINT USING " ###";QG(I,J);
810 NEXT J
820 LPRINT
830 NEXT I
840 NEXT PP
850 END
860 '
1000 ' =====
1010 ' == SUB 1000 ==
1020 ' =====
1030 DIM CN$(14) : RESTORE 1080
1040 FOR I=1 TO 14
1050 READ CN$(I)
1060 NEXT I
1070 '
1080 DATA t01,t02,t03,t04,t05,t06,t07
1090 DATA t08,t09,t10,t11,t12,t13,t14
1100 RETURN
1110 '
3000 ' =====
3010 ' == SUB 3000 ==
3020 ' =====
3030 REMOVE : MOUNT 2
3040 OPEN "2:"+A$ FOR INPUT AS #1
3050 INPUT #1,NO
3060 FOR IA=1 TO NO
3070 INPUT #1,QN(IA),NM$(IA)

```

```

3080 NEXT IA
3090
3100 FOR IA=1 TO NO
3110   FOR IJ=1 TO 9
3120     INPUT #1,Q(IA,IJ)
3130   NEXT IJ,IA
3140 CLOSE #1 : REMOVE
3150 RETURN

```

```

SUBROUTINE DOS(MD,NO)
DIMENSION MD(700,3),KD(10),NP(5,100),LD(10)
DATA IT/1H./,IA/1H*/,IB/1H /
DO 10 I=1,10
10 KD(I)=0
DO 20 J=1,NO
N=MD(J,2)/10
IF(N.EQ.10)N=9
KD(N+1)=KD(N+1)+1
20 CONTINUE
DO 30 I=1,10
30 LD(I)=I*10
WRITE(6,600)(LD(I),I=1,10)
600 FORMAT(1H1,20X,10I10)
DO 50 I=1,10
DO 60 J=1,5
DO 70 K=1,100
NP(J,K)=IB
IF(MOD(K,5).EQ.0)NP(J,K)=IT
70 CONTINUE
60 CONTINUE
M=KD(I)
IF(M.LE.500)GO TO 76
75 STOP
76 DO 80 J=1,M
IF(J.LE.100)NP(1,J)=IA
IF(J.GT.100.AND.J.LE.200)NP(2,J-100)=IA
IF(J.GT.200.AND.J.LE.300)NP(3,J-200)=IA
IF(J.GT.300.AND.J.LE.400)NP(4,J-300)=IA
IF(J.GT.400.AND.J.LE.500)NP(5,J-400)=IA
80 CONTINUE
KK=I*10-1
DO 90 J=1,5
IF(J.EQ.3)WRITE(6,601)KK
WRITE(6,602)(NP(J,K),K=1,100)
90 CONTINUE
WRITE(6,603)
50 CONTINUE
601 FORMAT(1H+,I3,I7)
602 FORMAT(1H ,20X,100A1)
603 FORMAT(1H )
RETURN
END

```


『組織内ソフトとパーソナルソフトの有効利用について』の中に落丁がありますので訂正致します。

P33のSUBプログラムはP22のプログラム名“check”の続きです。

P32のFORTRANプログラムは、抹消し下記のプログラムを挿入して下さい。

```
100 '
110 ' | seisek (1983年 9月) |
120 '
122 PRINT " |
124 PRINT " | seisek |
126 PRINT " | : for i=1 to 200 : next i
130 DEFINT Q
140 '
150 GOSUB 1000
160 '
170 DIM QN(50),NM$(50),Q(50,9)
180 PRINT CHR$(12)
190 INPUT"ナンクミ ノ セイセキ テ`スカ";K
200 A$=CN$(K)
210 '
220 GOSUB 3000
230 '
235 PRINT
240 INPUT"カモク コク=1 シャ=2 スウ=3 リカ=4 エイ=5" ;M
245 PRINT
250 FOR I=1 TO NO
260 PRINT USING "####: & &";QN(I),NM$(I); : INPUT Q(I,M)
270 NEXT I
280 BEEP
290 '
300 GOSUB 2000
310 '
315 PRINT
320 INPUT "ホカ ノ カモク ラ input シマスカ? yes=1 no=2";Y
330 IF Y=1 THEN 240
340 '
345 PRINT
350 INPUT "ホカ ノ クラスノ セイセキ ラ ニュウリョク シマスカ? yes=1 no=2";Y
360 IF Y=1 THEN 180
370 END
```

```

380 '
1000 ' =====
1010 ' ==      SUB  1000      ==
1020 ' =====
1030 DIM CN$(14)
1040 FOR I=1 TO 14
1050   READ CN$(I)
1060 NEXT I
1070 '
1080 DATA t01,t02,t03,t04,t05,t06,t07
1090 DATA t08,t09,t10,t11,t12,t13,t14
1100 RETURN
1110 '
2000 ' =====
2010 ' ==      SUB  2000      ==
2020 ' =====
2030 REMOVE : MOUNT 2
2040 OPEN "2:"+A$ FOR OUTPUT AS #1
2050   PRINT #1,NO
2060   FOR II=1 TO NO
2070     PRINT #1,QN(II),NM$(II)
2080   NEXT II
2090 '
2100   FOR II=1 TO NO
2110     FOR IJ=1 TO 9
2120       PRINT #1,Q(II,IJ)
2130     NEXT IJ,II
2140 CLOSE #1 : REMOVE
2150 RETURN
2160 '
3000 ' =====
3010 ' ==      SUB  3000      ==
3020 ' =====
3030 REMOVE : MOUNT 2
3040 OPEN "2:"+A$ FOR INPUT AS #1
3050   INPUT #1,NO
3060   FOR II=1 TO NO
3070     INPUT #1,QN(II),NM$(II)
3080   NEXT II
3090 '
3100   FOR II=1 TO NO
3110     FOR IJ=1 TO 9
3120       INPUT #1,Q(II,IJ)
3130     NEXT IJ,II
3140 CLOSE #1 : REMOVE
3150 RETURN

```

```

7010 ' ==      midashi uchidaşi      ===
7020 ' =====
7030 '
7040 RESTORE 7310
7050 LPRINT TAB(15);
7060 FOR JM=1 TO 5
7070 LPRINT "  ";
7080 GOSUB 7210
7090 NEXT JM
7100 '
7110 LPRINT "  3";
7120 GOSUB 7210
7130 LPRINT "  5";
7140 GOSUB 7210
7150 FOR JM=1 TO 2
7160 LPRINT "  ";
7170 GOSUB 7210
7180 NEXT JM
7190 LPRINT : LPRINT
7200 RETURN
7210 LPRINT CHR$(&H1B);"I";"0015":
7220 FOR IM=1 TO 30
7230 READ D$ : C=VAL("&h"+D$) : GOSUB 7270
7240 NEXT IM
7250 RETURN
7260 '
7270 '**** PUT CHARACTORE      for PC-8801 ****
7280 A=INP(&H40) AND 1 : IF A<>0 THEN 7280
7290 OUT &H10,C : OUT &H40,0 : OUT &H40,1 : RETURN
7300 '
7310 ' コク
7320 DATA 00,00,fe,ff,02,80,02,90,12,91,12,91,12,91,f2,9f
7330 DATA 12,91,12,91,12,95,12,99,02,90,fe,ff,00,00
7340 ' シャ
7350 DATA 20,08,22,04,22,02,24,ff,24,03,a8,04,68,08,20,40
7360 DATA 00,40,40,40,40,40,ff,7f,40,40,40,40,00,40
7370 ' スウ
7380 DATA 49,84,2a,5c,08,27,ff,64,08,94,2a,0c,49,04,20,80
7390 DATA 1f,40,08,30,c8,0c,08,03,c8,08,28,30,08,c0
7400 ' リ
7410 DATA 00,00,42,20,42,20,fe,3f,42,10,42,08,42,08,00,20
7420 DATA 3e,21,2a,21,2a,21,fe,3f,2a,21,2a,21,3e,20
7430 ' エイ
7440 DATA 00,82,04,42,04,22,ef,23,24,12,24,12,24,0a,e4,07
7450 DATA 24,0a,24,12,2c,12,e7,23,04,22,04,42,00,82
7460 ' ケイ
7470 DATA 04,00,24,f9,24,89,27,89,24,f9,04,00,44,00,40,00
7480 DATA 40,00,40,00,ff,ff,40,00,40,00,40,00,40,00
7490 ' ケイ
7500 DATA 04,00,24,f9,24,89,27,89,24,f9,04,00,44,00,40,00
7510 DATA 40,00,40,00,ff,ff,40,00,40,00,40,00,40,00
7520 ' ヘン
7530 DATA 00,03,c0,00,f8,ff,07,04,80,03,78,ff,49,09,49,09
7540 DATA 49,ff,49,09,49,09,49,ff,49,09,49,49,70,ff
7550 ' ショ
7560 DATA 00,c0,00,30,fc,0f,04,00,14,02,14,02,14,22,57,42
7570 DATA 94,fe,54,03,34,02,14,0a,04,06,00,02,00,00
7580 END

```

```

100 '
110 ' | clear (1983年 9月) |
120 '
121 PRINT CHR$(12)
122 PRINT " "
124 PRINT "| clear |"
126 PRINT " "
128 '
130 ON ERROR GOTO 330
140 DEFINT Q
150 '
160 GOSUB 1000
170 '
180 DIM QN(50),NM$(50),Q(50,9)
190 '
200 PRINT : INPUT "セイセキヲ clear スル クラス";K
210 A$=CN$(K)
220 REMOVE : MOUNT 2
230 OPEN "2:"+A$ FOR INPUT AS #1
240 INPUT #1,NO
250 FOR I=1 TO NO
260 INPUT #1,QN(I),NM$(I)
270 NEXT I
280 CLOSE #1 : REMOVE
290 '
300 GOSUB 2000
310 '
320 GOTO 340
325 '
330 PRINT : PRINT K;" クミ ノ メイホ` ヤリナオシ"
340 ON ERROR GOTO 0
350 END
360 '
1000 ' =====
1010 ' == SUB 1000 ==
1020 ' =====
1030 DIM CN$(14)
1040 FOR I=1 TO 14
1050 READ CN$(I)
1060 NEXT I
1070 '
1080 DATA t01,t02,t03,t04,t05,t06,t07
1090 DATA t08,t09,t10,t11,t12,t13,t14
1100 RETURN
1110 '
2000 ' =====
2010 ' == SUB 2000 ==
2020 ' =====
2030 REMOVE : MOUNT 2
2040 OPEN "2:"+A$ FOR OUTPUT AS #1
2050 PRINT #1,NO
2060 FOR II=1 TO NO
2070 PRINT #1,QN(II),NM$(II)
2080 NEXT II
2090 '
2100 FOR II=1 TO NO
2110 FOR IJ=1 TO 9
2120 PRINT #1,Q(II,IJ)
2130 NEXT IJ,II

```

2140 CLOSE #1 : REMOVE
2150 RETURN

Fig.12 A iexample of output

1 クミ 44 ニン

	国	社	数	理	英	3計	5計	偏	序			
101	ア	ト	ウ	26	22	0	28	12	38	88	37	555
102	オ	カ	タ	83	97	87	72	93	263	432	66	26
104	カ	ケ	フ	84	84	63	92	93	240	416	65	45
105	カ	サ	マ	46	72	27	42	17	90	204	47	360
106	キ	タ	ム	61	96	30	56	58	149	301	55	206
107	ク	ト	ウ	42	33	39	58	15	96	187	45	400
108	サ	ノ		50	78	39	54	34	123	255	51	272
109	ナ	カ	オ	59	71	57	68	60	176	315	56	184
110	マ	ユ	ミ	45	59	24	46	24	93	198	46	378

ソウスウ 620 ニン

カモクハツ ハイキン

クミ	コクコ	シャカイ	スウカク	リカ	エイコ	3ケイ	5ケイ
1	60.0	67.4	35.1	51.7	46.3	141.4	260.4
2	54.6	58.0	30.5	46.7	41.7	126.8	231.5
3	52.4	66.9	33.3	48.6	44.2	129.9	245.4
4	55.7	30.9	61.5	50.0	40.7	157.9	238.8
5	54.9	61.1	27.0	45.6	32.3	114.2	220.9
6	50.0	56.5	34.4	43.6	39.9	124.3	224.5
7	57.3	61.0	33.6	47.7	43.4	134.3	243.1
8	58.7	63.7	34.5	49.3	39.9	133.1	246.1
9	54.3	58.9	34.8	52.5	39.2	128.3	239.8
10	59.2	60.7	33.9	48.9	37.3	130.3	240.0
11	53.8	57.7	32.5	46.0	41.6	127.8	231.5
12	56.2	57.7	32.8	45.5	40.3	129.3	232.5
13	57.5	62.3	33.0	49.8	37.6	128.1	240.1
14	56.3	58.7	34.9	53.0	38.1	129.3	240.9
TT	55.8	58.7	35.1	48.5	40.2	131.1	238.2

ファンフ

	- 9	-19	-29	-39	-49	-59	-69	-79	-89	-100	(マデ)
コクコ	---	33	12	53	61	66	87	87	108	85	28
シャカイ	---	41	19	51	59	54	70	63	78	77	108
スウカク	---	124	64	83	110	64	57	53	28	19	18
リカ	---	30	25	80	98	93	78	73	75	55	13
エイコ	---	112	95	72	59	51	44	48	48	53	38
3ケイ	---	48	75	78	90	81	65	72	59	44	8
5ケイ	---	34	46	81	83	90	77	74	74	50	11

4. 有効利用の効果について

一つのシステムの効果を判定するとき、初期の段階では直接効果、間接効果がよく論じられた。すなわち前者は担当者とのチャージ比較であって、金額に換算できる直接的効果である。

後者は信頼性、種々の検討、担当者の空き時間の活用等、直接金銭に結びつかない間接的な効果である。

人の手でやっていた仕事を電子計算機におき替えることが主体である頃の話である。

現在の情報化時代においては組織の中に計算機があるのではなく、計算機を中心に組織が構成されているといっても過言ではない。このような状況下で作成される組織内ソフトに対して、業務の目的によって規格化されたデータバンクから利用者に適切な情報を提供する種々のシステムが要求されるのは当然と言える。

一方業務の多用化と小型電子計算機の普及は、さらに組織内ソフトを充実する促進剤となり、ますます高度情報社会へのテンポを早めている。このような観点に立って組織内ソフトとパーソナルソフトの効果を論じれば、組織効果と展開効果を提案したい。組織効果とは一連のデータバンクを人手を介することなく、関連部門に適切な情報をいかに提供できるかということであり、展開効果とは必要に応じて新たな情報に加工できる効果である。前者は組織（特に管理者）の能力を要求し、後者は利用者（担当者）の能力を必要とする。その基本は、組織内ソフトのデータを作成するに当って、そのデータを最大限に活用する入力手段を構想することであると考えられる。その意味で、前節のプログラム例は低レベルの有効利用である。しかし大型機と接続していない現実のもとでは、OCRとコード変換システムの利用効果は非常に大である。

5. あとがき

近年、小型電子計算機用の I/O デバイスが種々開発され、入力手段も多様化している。このことはパーソナルソフトによる一次データの加工を容易にし、大型機用のソフトあるいは組織内ソフトをさらに充実させるものと考えられる。当面の有効利用として、パソコン用、マークカード読取機、XYプロッタ、座標読取機の利用を進めているが、音声入力も興味ある入力手段であり、その利用も検討したい。有効利用とその効果は機器構成と組織のあり方によって左右され、情報化の進展につれ、あらゆる分野で改変がなされるものと考えられる。

最後に、今回のソフト作成、テストランに際し C A I 研究会のメンバーである東大阪市立繩手中学校・西本宗成、藤田裕司、前田浩、高槻市立高槻第九中学校・西村三夫、宝塚市立宝塚第一中学校・藤井晴基の諸先生方のご協力に感謝いたします。